COMPUTER PROGRAM FOR STOCHASTIC UTILIZATION OF THE USLE



Southern Piedmont Conservation Research Center
Agricultural Research Service, USDA
Watkinsville, GA 30677

RESEARCH REPORT 28
No. IRC 050189

May 1989

United States
Department of
Agriculture

National Agricultural Library

Research Report[1]
No. IRC 050189

Computer Program for Stochastic Utilization of the USLE

A. W. Thomas
Agricultural Engineer
Southern Piedmont Conservation Research Center
Agricultural Research Service, USDA
Watkinsville, GA 30677

W. M. Snyder
Hydrologic Engineer
275 Gatewood Circle
Athens, GA 30607

and

A. L. Dillard
Agricultural Engineer
Southern Piedmont Conservation Research Center
Agricultural Research Service, USDA
Watkinsville, GA 30677

May 1989

[1]/ This report describes a computer program and its operational details for randomizing two factors of the Universal Soil Loss Equation and then simulating both seasonal pattern and annual total soil loss for a selected agronomic practice.

PREFACE

This report is one of a series of "Research Reports" published by the Integrated Row Crop Management Systems Research Unit at the Southern Piedmont Conservation Research Center, Watkinsville, GA. The purpose of these reports is to provide a mechanism for technology transfer to potential users of information developed by the scientists in this Research Unit. Below is a list of the reports developed in this series. Any report may be obtained by requesting it at the Watkinsville address (phone 404/769-5631 or FTS 250-2425).

Research Report No.	Authors and Title
IRC 060183	White, A. W., Jr., R. R. Bruce, A. W. Thomas, G. W. Langdale, and H. F. Perkins. The effects of soil erosion on soybean production in the Southern Piedmont of Georgia in 1982.
IRC 093083	Welch, R., T. R. Jordan, A. W. Thomas, and J. W. Ellis. Photogrammetric techniques for monitoring soil erosion.
IRC 060184	White, A. W., Jr., R. R. Bruce, A. W. Thomas, G. W. Langdale, and H. F. Perkins. Effect of soil erosion on soybean yields and characteristics of Cecil-Pacolet soils.
IRC 070184	Thomas, A. W. and W. M. Snyder. Computer programs for analysis and simulation of probability distribution using sliding polynomials.
IRC 010186	Thomas, A. W., W. M. Snyder, and A. L. Dillard. A computer program for transforming stochastic data and evaluating probability distributions.
IRC 060686	Harper, L. A. and W. M. Snyder. A laboratory guide to smoothing and use of stochastic integrals for time-distribution data (1st edition 6/6/86, 2nd edition 12/1/87).
IRC 070686	Harper, L. A. and W. M. Snyder. A laboratory guide to sliding polynomial smoothing and testing for significant difference of functions in non-standard data sets (1st edition 7/6/86, 2nd edition 5/1/88).
IRC 070187	Thomas, A. W., W. M. Snyder, and A. L. Dillard. Computer programs for analysis and simulation of seasonally continuous probability distributions.

IRC 040188	Thomas, A. W., W. M. Snyder, and A. L. Dillard. Computer programs for a form-free bivariate model -- Analysis and Simulation --.
IRC 050189	Thomas, A. W., W. M. Snyder, and A. L. Dillard. Computer program for stochastic utilization of the USLE.

INTRODUCTION

In systems involving crop production and soil and water conservation, planners and producers are interested in both magnitudes and seasonal variations of climatological variables such as rainfall and temperature. Because of limited information for planning purposes, we have prepared a series of papers that is intended to provide a class of utilitarian methodologies to assist farm planners and conservationists in assessing stochastic impacts of variables affecting their decisions and operations.

The series of papers was developed under the general theme of "Stochastic Impacts on Farming". Companion research reports were prepared to provide the user the computer programs and technology for utilizing the new methodologies. In the first paper, Thomas and Snyder (1986a) developed a data transformation method for climatological records that was required for following analyses. The companion research report by Thomas et al. (1986) described the computer program and technique for transforming the stochastic data. In the second paper, Snyder and Thomas (1986) presented a method for computing seasonally continuous probability distributions. A third paper (Thomas and Snyder, 1986b) used simulation techniques to develop seasonal variation of risk and uncertainty. A research report (Thomas et al., 1987) provided the computer programs and their operational details used in the studies of paper two (analysis) and paper three (simulation). A fourth paper (Snyder and Thomas, 1987) presented a method for generating synthetic inputs from bivariate distributions that may be used in the modeling of agricultural systems. A companion research report (Thomas et al., 1988a) provided the computer programs and procedures for fitting

a two-dimensional sliding polynomial surface to a bivariate distribution and then simulating data pairs from this smoothed probability surface.

The fifth and last paper (Thomas et al. 1988b) in this series illustrated practical application of seasonal risk analysis for soil conservation planning. Building on the background of the earlier papers we demonstrated applicability of stochastic analysis by quantifying potential reduction in soil loss through seasonal adjustment of agricultural practice. Reduction in soil loss was expressed in terms of expectation (risk) and in terms of variability of that expectation (uncertainty).

This research report provides the computer program and its operational details for randomizing two factors of the Universal Soil Loss Equation and then simulating both seasonal pattern and annual total soil loss for a selected agronomic practice.

METHODOLOGY

The Universal Soil Loss Equation is used to demonstrate our stochastic approach in evaluating soil loss risk for conservation planning. The equation is:

$$A = RCKLSP \qquad (1)$$

where A is the computed soil loss per unit area, R is the rainfall and runoff factor, C is the cover and management factor, K is the soil erodibility factor, L is the slope-length factor, S is the slope-steepness factor, and P is the support practice factor.

To demonstrate the stochastic analysis of soil loss, we simulate randomized variates, namely, factors "R" and "C". First, we equate "R" to EI, where EI is the erosion index (Wishchmeier and Smith, 1978). Next, we define a standardized area where the factors K, L, S, and P produce a product W. For convenience, we set a value of 0.01 for W to define the standardized area and produce reasonable values of soil loss in our example. Then we may write:

$$A(s)_m = EI_m C_m W \quad (2)$$

where $A(s)_m$ is the soil loss for month m from the standardized area, EI_m is the erosion index for that month, and C_m is the cover and management factor for the same month. C_m is equated to the appropriate cropstage soil loss ratio by month (Wischmeier and Smith, 1978). We divide equation (1) expressed for a month by equation (2) to obtain:

$$A_m = A(s)_m \; KLSP/W \quad (3)$$

where A_m is the soil loss for a month m for a real area defined by KLSP. The use of the randomized EI_m and C_m variates thus demonstrates the stochastic approach to potential soil loss for a standardized area that can be readily converted to a real area.

Randomization of the EI_m term follows directly from earlier work by Snyder and Thomas (1986) and Thomas and Snyder (1986b) where continuous monthly probabilities of EI were calculated. In simulation, computer-generated random numbers in the scale of 0 to 100 were considered probabilities, and those probabilities were converted to

random EI values using the calculated monthly probability distributions (Snyder and Thomas, 1986; Fig. 4). Form-free sliding polynomials used in the previous work for estimating parent distributions, yielded estimates of system variance. This system variance is included in the simulation by randomly pulsing the nodes of the polynomials prior to conversion of the random probabilities to random EI values (Thomas and Snyder, 1986b).

Randomization of the C_m term is described by Thomas et al. (1988). They defined a bounded double-triangle distribution and calibrated it against a bounded normal distribution (Thomas et al., 1988; Fig. 2). In simulation computer-generated random numbers in the scale of 0 to 100 were considered probabilities, and those probabilities were converted to random C_m values using the calibrated double-triangle distribution.

In brief summary, the methodology is as follows. The nodes of the seasonal EI distribution are pulsed with random normal numbers using standard deviations derived during the probability analysis (Snyder and Thomas, 1986). This pulsing incorporates the system variance. Next, two random numbers are generated. One random number is converted to an EI_m value using the seasonal probability distributions determined earlier by Snyder and Thomas (1986). The second random number is converted to a C_m value with the bounded double triangle distribution. The product of W, the randomized EI_m and C_m is a randomized $A(s)_m$ value. In the example of this report values of $A(s)_m$ for each month for a ten-year planning period were generated. This process was repeated until 100 ten-year periods were generated.

DISCUSSION

Mono-cropped soybeans was selected as the crop for demonstration purposes of this research report. The selected practices of conventional-till and fair-productivity were used as defined in Table 5 of Wischmeier and Smith (1978). Late-season planting in this case was assumed to be on June 15. Based on these definitions, C values were selected from the table for each of the cropstage periods. Next, dates bracketing each cropstage were assigned. The dates were assigned to reflect general agronomic practices of the Piedmont physiographic region in the Southeast. Cropstage periods were required to begin or end on the first or the middle of the month. For those months with a cropstage change in the middle of the month, the use of the first-half or second-half C value was randomized. The resulting effect is to produce more variability in soil loss at the beginning or ending of cropstage periods. The resulting mean values of the 24 C-factors for this selected cropping system are shown in part "a" of Fig. 1.

Monthly values of potential soil loss were generated for each month for 100 ten-year planning periods for the mono-cropped soybeans. The following procedures illustrate the calculations. After a set of ten-year values is simulated, the monthly values are arranged in order of magnitude; the ten January values are ordered, then the ten February values, and so on. After 100 ten-year sets are simulated, the values for each rank number from 1 to 10 are further arranged in order of magnitude from 1 to 100. From the ordered set of 1 to 100, we can select, say number 5. We then have a value for the expected soil loss in January that will be equalled or exceeded once in a ten-year planning period, and we are 95% confident that the true value will be smaller

than this selected 1-in-10 value. This procedure is repeated for each calendar month and for each risk level of expected number of occurrences in ten years.

Seasonal risks of soil loss are computed for the fair-productivity, late-season, conventional-till soybeans. These simulated results, generated at the 95% confidence level, are shown in part "b" of Fig. 1. For example, the 1-in-10 risk levels of soil loss peak with a value near 3.25 Mg/ha in July and drop to about 0.3 Mg/ha in October. Twice in the ten-year planning period, we would expect soil loss equal to or greater than about 2.4 Mg/ha in July. Due to the large C-factor and probability of a large EI value during July, the expected soil loss peaks in July at all risk levels.

In addition to the monthly soil loss rates, the annual total soil loss is also computed. The annual totals are ranked from one to ten and then each of these ranks is ordered from 1 to 100. Number 5 in this list, as an example, is an estimate of the 95% confidence level for each annual risk of 1-in-10, 2-in-10, and so on. The 95% confidence levels of the annual soil loss risks for the monocropped soybeans are given in Table 1.

REFERENCES

1. Snyder, W. M. and A. W. Thomas. 1986. Stochastic Impacts on Farming: II. Seasonally Continuous Probability Distributions. Trans. ASAE. 29(4):1017-1025.

2. Snyder, W. M. and A. W. Thomas. 1987. Stochastic Impacts on Farming: IV. A Form-Free Bivariate Distribution to Generate Inputs to Agricultural Models. Trans. ASAE. 30(4):946-952.

3. Thomas, A. W. and W. M. Snyder. 1986a. Stochastic Impacts on Farming: I. Transformation of Agriculturally Relevant Variates for Probability Analysis. Trans. ASAE. 29(1):128-134.

4. Thomas, A. W. and W. M. Snyder. 1986b. Stochastic Impacts on Farming: III. Simulation of Seasonal Variation of Climatic Risk. Trans. ASAE. 29(4):1026-1031.

5. Thomas, A. W., W. M. Snyder, and A. L. Dillard. 1986. A Computer Program for Transforming Stochastic Data and Evaluating Probability Distributions. Technical Research Report No. IRC 010186. Watkinsville, GA.

6. Thomas, A. W., W. M. Snyder, and A. L. Dillard. 1987. Computer Programs for Analysis and Simulation of Seasonally Continuous Probability Distributions. Technical Research Report No. IRC 070187. Watkinsville, GA.

7. Thomas, A. W., W. M. Snyder, and A. L. Dillard. 1988a. Computer Programs for a Form-Free Bivariate Model -- Analysis and Simulation --. Technical Research Report No. IRC 040188. Watkinsville, GA.

8. Thomas, A. W., W. M. Snyder, and G. W. Langdale. 1988b. Stochastic Impacts on Farming: V. Risk Adjustment Through Conservation Planning. Trans. ASAE. 31(5):1368-1374.

9. Wischmeier, W. H. and D. D. Smith. 1978. Predicting Rainfall Erosion Losses - A Guide to Conservation Planning. USDA Agriculture Handbook 537, U. S. Government Printing Office, Washington, D. C. 58 pp.

TABLE 1. SIMULATED ANNUAL SOIL LOSS FOR MONO-CROPPED SOYBEANS

Soybean Practice	Annual soil loss, Mg/ha — Risk: Rank Numbers in Ten				
	1	2	4	7	10
Conventional-Till Fair-Productivity Late-Season	6.0	5.0	4.0	3.0	2.3



Fig. 1 - Patterns of C factors and risk levels of soil loss for conventional-till, fair productivity, late-season soybeans.

APPENDIX

The Appendix includes a description of input and output variables, program listing, and sample output for randomizing the EI and C factors of the Universal Soil Loss Equation and then simulating both seasonal pattern and annual total soil loss for a selected agronomic practice. Seasonal risks of soil loss are shown for fair-productivity, late-season, conventional-till, mono-cropped soybeans.

The program is presented for the convenience of potential users. While the program has been run and tested on various data sets, the originators of the program assume no responsibility for its accuracy or adequacy. Such responsibility must rest solely on the user. We stand ready to assist and advise within the limitations imposed by our operating resources. The program is listed in FORTRAN 77.
(Trade name is included for the benefit of the reader and does not imply an endorsement or preferential treatment of the named product.)

Program Inputs

Variable Name	Comment
AT(I)	Problem title. For I=1 to 80
COEF(I)*	Sliding polynomial coefficients. For I=1 to 4
NR,N1,N2	Random number seeds (Range of 0 to 30000)
NI	Number of planning periods to be simulated.
NP	Length of planning period in years.
IPRIN	Print option control Program prints information in sample output. Values of IPRIN and options available are IPRIN Description 1 or 2. By NI; monthly values and yearly totals unranked by planning period. 1 or 3. By NI; monthly values and yearly totals ranked by planning period. 4. No additional output.
HB(I) HSD(I) HSK(I) HMIN(I)	Monthly mean of EI. Monthly standard deviation of EI. Monthly coeff. of skew of EI. Monthly minimum class limit of EI. For I=1 to 12
YN(I,J) SN(I,J)	Sliding polynomial solution nodes Nodal standard deviation. For I=2 to 5: J=3 to 7 Values of the above six variables obtained from probability analysis of monthly EI data (Thomas et al., 1987).
CF(I)	Cropping factors in the USLE by half-months (Fig. 1 part a). For I=1 to 24

*COEF (1) = -0.074074
COEF (2) = 0.777817
COEF (3) = 0.333288
COEF (4) = -0.037032

Program Outputs

Variable name	Comment
AT(I)	Problem title. For I=1 to 80
NR,N1,N2	Random number seeds.
NI	Number of planning periods to be simulated.
NP	Length of planning period in years.
HB(I)	Monthly mean of EI.
HSD(I)	Monthly standard deviation of EI.
HSK(I)	Monthly coeff. of skew of EI.
HMIN(I)	Monthly minimum class limit of EI. For I=1 to 12
YN(I,J)	Sliding polynomial solution nodes.
SN(I,J)	Nodal standard deviation. For I=2 to 5: J=3 to 7
CF(I)	Cropping factors in USLE. For I=1 to 24
HP(I)	Minimum value of intermediate transform variate, h', at minimum class limit
CP(I)	Common point of two exponential transform limbs (in transform scale).
VR(I)	Right asymptotic boundary for transform limb v2 (in transform scale).
F1(I)	Transform shape parameter.
D1(I)	Transform shape parameter. For I=1 to 12
ER(M,K,N)	Monthly value of erosion (C*EI). For K=1 to NP: M=1 to NI: N=1 to 12
ERY(M,J)	Yearly totals of erosion. For J=1 to NP: M=1 to NI

Program Listing

```
C#####################################################################
C#####             'EROSION.FOR'                            ###########
C#####  CALCULATE SOIL LOSS FOR UNITIZED AREA BY USING      ###########
C#####  THE PRODUCT OF MONTHLY EI AND MONTHLY CROPPING      ###########
C#####  FACTOR.  USES SEASONALLY CONTINUOUS PROBABILITY     ###########
C#####  DISTRIBUTION OF EI AND A DOUBLE TRIANGLE DIST-      ###########
C#####  RIBUTION FOR CROPPING FACTOR TO OBTAIN MONTHLY      ###########
C#####  AND YEARLY DISTRIBUTIONS OF SOIL LOSS.              ###########
C#####################################################################
      DIMENSION ER(100,20,12), ERY(100,20), AT(80), HP(12)
      DIMENSION RMAX(100,20,2)
      COMMON /INT/ YN(7,7),SN(7,7),COEF(4),SNOD(12,7),YNOD(12,7)
      COMMON /EI/ HSK(12), RNOD(12,7), HB(12), HSD(12), HMIN(12),
     *    CP(12), VR(12), F1(12), D1(12), C(4), EIC
      COMMON /CRP/ CF(24), CC
      COMMON /RNN/ NR, N1, N2
      OPEN(5,FILE='G2EI')
      OPEN(6,FILE='EROSION.OUT',STATUS='NEW')
      WRITE(*,2039)
      READ(7,1003) (AT(I),I=1,80)
      WRITE(6,2001) (AT(I),I=1,80)
      READ(5,1001) (COEF(I),I=1,4)
      READ(5,1005) NR,N1,N2,NI,NP,IPRIN
      WRITE(6,2007) NR,N1,N2,NI,NP
      WRITE(6,2009)
      READ(5,1006) (HB(I),HSD(I),HSK(I),HMIN(I),I=1,12)
      WRITE(6,2011) (I,HB(I),HSD(I),HSK(I),HMIN(I),I=1,12)
      KNT=1
      DO 10 I=2,5
      DO 5 J=3,7
      READ(5,1007) YN(I,J),SN(I,J)
      YNOD(KNT,J)=YN(I,J)
      SNOD(KNT,J)=SN(I,J)
    5 CONTINUE
      KNT=KNT+3
      WRITE(6,2015) (YN(I,J),SN(I,J),J=3,7)
   10 CONTINUE
      READ(7,1009) (CF(I), I=1, 24)
      WRITE(6,2016) (CF(I), I=1, 24)
      CALL INTERP
      WRITE(6,2017)
      DO 15 I=1,12
      HP(I)=(HMIN(I)-HB(I))/HSD(I)+HSK(I)/8
      CP(I)=2.+0.375*HSK(I)
      VR(I)=4.05+(CP(I)-2.)
      F1(I)=ALOG((4.-VR(I))/(CP(I)-VR(I)))/HP(I)
      D1(I)=F1(I)*(VR(I)-CP(I))/CP(I)
      WRITE(6,2019) HP(I),CP(I),VR(I),F1(I),D1(I)
      DO 15 L=1,7
      RNOD(I,L)=YNOD(I,L)
   15 CONTINUE
      CALL RAND(R,1)
```

```
C######################################################################
C#####   BEGIN LOOPS                                        ###########
C#####   K = MONTH                                          ###########
C#####   I = PLANNING PERIOD                                ###########
C#####   M = NUMBER OF SAMPLES (SIMULATED PERIODS)          ###########
C######################################################################
      DO 115 M=1,NI
      DO 70 I=1,NP
      DO 60 K=1,12
      CALL EICAL(K)
      CALL CFAC(K)
      ER(M,I,K) = EIC * CC
   60 CONTINUE
C######################################################################
C#####   END OF MONTH LOOP. CALCULATE YEARLY TOTAL, THE     ###########
C#####   MONTH WITH THE LARGEST SOIL LOSS AND THE           ###########
C#####   PERCENTAGE OF THE YEARLY TOTAL OF THE LARGEST      ###########
C#####   MONTH.                                             ###########
C######################################################################
      DO 65 K=1, 12
      ERY(M,I) = ERY(M,I) + ER(M,I,K)
   65 CONTINUE
      RMAX(M,I,1) = 0.
      DO 70 K=1, 12
      RM = ER(M,I,K)/ERY(M,I)
      IF (RM.LT.RMAX(M,I,1)) GO TO 70
      RMAX(M,I,1) = RM
      RMAX(M,I,2) = FLOAT(K)
   70 CONTINUE
C######################################################################
C#####   END OF PLANNING PERIOD. PRINT UNORDERED SAMPLES    ###########
C######################################################################
      GOTO (73,73,78,78), IPRIN
   73 CONTINUE
      WRITE(6,2030) M
      WRITE(6,2031) (K,K=1, 12)
      DO 75 I=1, NP
      WRITE(6,2033) I,(ER(M,I,K),K=1, 12),ERY(M,I),
     *              RMAX(M,I,1)*100., RMAX(M,I,2)
   75 CONTINUE
C######################################################################
C#####   RANK MONTHLY VALUES BY PLANNING PERIOD             ###########
C######################################################################
   78 CONTINUE
      DO 90 K=1,12
      DO 85 J=1,NP
      X6=ER(M,J,K)
      DO 80 J1=1,NP
      IF(ER(M,J1,K).GE.X6) GO TO 80
      ER(M,J,K)=ER(M,J1,K)
      ER(M,J1,K)=X6
      X6=ER(M,J,K)
   80 CONTINUE
   85 CONTINUE
   90 CONTINUE
```

```
C#####################################################################
C#####   RANK YEARLY TOTALS BY PLANNING PERIOD              ###########
C#####################################################################
      DO 100 J=1,NP
      X6=ERY(M,J)
      DO 95 J1=1,NP
      IF(ERY(M,J1).GE.X6) GO TO 95
      ERY(M,J)=ERY(M,J1)
      ERY(M,J1)=X6
      X6=ERY(M,J)
   95 CONTINUE
  100 CONTINUE
C#####################################################################
C#####             PRINT ORDERED SAMPLES                    ###########
C#####################################################################
      GOTO (103,110,103,110), IPRIN
  103 CONTINUE
      WRITE(6,2032) M
      WRITE(6,2028) (K,K=1, 12)
      DO 105 I=1, NP
      WRITE(6,2034) I,(ER(M,I,K),K=1, 12),ERY(M,I)
  105 CONTINUE
  110 CONTINUE
  115 CONTINUE
C#####################################################################
C#####           END OF SAMPLES                             ###########
C#####################################################################
C#####   RANK MONTHLY VALUES AND YEARLY TOTALS BY SAMPLE    ###########
C#####   ACROSS PLANNING PERIOD                             ###########
C#####################################################################
      DO 140 J1=1,12
      DO 135 K=1,NP
      DO 130 I=1,NI
      CHECK=ER(I,K,J1)
      DO 125 J=1,NI
      IF(ER(J,K,J1).GE.CHECK) GO TO 125
      ER(I,K,J1)=ER(J,K,J1)
      ER(J,K,J1)=CHECK
      CHECK=ER(I,K,J1)
  125 CONTINUE
  130 CONTINUE
  135 CONTINUE
  140 CONTINUE
      DO 155 K=1,NP
      DO 150 I=1,NI
      CHECK=ERY(I,K)
      DO 145 J=1,NI
      IF(ERY(J,K).GE.CHECK) GO TO 145
      ERY(I,K)=ERY(J,K)
      ERY(J,K)=CHECK
      CHECK=ERY(I,K)
  145 CONTINUE
  150 CONTINUE
  155 CONTINUE
```

```
C#####################################################################
C#####   PRINT RANKED VALUES BY MONTH                       ###########
C#####################################################################
      WRITE(6,2025)
      DO 170 N=1,12
      WRITE(6,2021) N
      WRITE(6,2027) (L,L=1,NP)
      DO 165 M=1,NI
      WRITE(6,2029) M,(ER(M,K,N),K=1,NP)
  165 CONTINUE
  170 CONTINUE
C#####################################################################
C#####   PRINT RANKED YEARLY TOTALS                         ###########
C#####################################################################
      WRITE(6,2035)
      WRITE(6,2027) (L,L=1, NP)
      DO 180 M=1, NI
      WRITE(6,2037) M, (ERY(M,J), J=1, NP)
  180 CONTINUE
  700 STOP
 1001 FORMAT(4F12.0)
 1003 FORMAT(80A1)
 1005 FORMAT(3I5,2I10,I5)
 1006 FORMAT(4F10.0)
 1007 FORMAT(2F10.0)
 1009 FORMAT(8F10.0)
 2001 FORMAT(' ',80A1)
 2005 FORMAT(' '/,20X,'INPUT DATA'/,9X,'SEEDS',9X,'NI',9X,'NP')
 2007 FORMAT(' ',6X/,5I8)
 2009 FORMAT(' '/,' MONTH   MEAN  STD DEV    SKEW   HMIN')
 2011 FORMAT(' ',I4,2F12.3,2F12.3)  2013 FORMAT(' '/,10X,'SAMPLE NO.',I4)
 2015 FORMAT(' ',2F12.3)
 2016 FORMAT(' ', 'CROPPING FACTORS'/,3(8F10.2/))
 2017 FORMAT(' ',8X,'HP',8X,'CP',8X,'VR',8X,'F',9X,'D'/)
 2019 FORMAT(' ',5F10.4)
 2021 FORMAT(' '/,10X,'MONTH',I4)
 2023 FORMAT(' ',8(I4,F8.3)/)
 2025 FORMAT(' '/, ' MONTHLY TOTALS '/)
 2027 FORMAT(' ',//11X,'  SORTED DATA '/,9X,20(I2,5X))
 2028 FORMAT('  ', 12(I8), '   TOTAL')
 2029 FORMAT(' ',I3,2X,20F7.2)
 2030 FORMAT(' '/, ' UNORDERED SAMPLES NUMBER ',I5/)
 2031 FORMAT('  ', 12(I8), '   TOTAL', 3X, ' %   AT MONTH')
 2032 FORMAT(' '/, ' ORDERED SAMPLES NUMBER ',I5/)
 2033 FORMAT(' ', I3, 12F8.2, 2F8.2, F4.0)
 2034 FORMAT(' ', I3, 12F8.2, F8.2)
 2035 FORMAT(' '/, ' YEARLY TOTALS '/)
 2037 FORMAT(' ',I3,2X,20F7.2)
 2039 FORMAT(' ', ' UNIT 7 FOR C FACTOR VALUES '/)
      END
C#####################################################################
C#####   SUBROUTINE INTERP                                  ###########
C#####   INTERPOLATE MONTHLY DISTRIBUTIONS OF EI FROM       ###########
C#####   SLIDING POLYNOMIAL SOLUTION NODES.                 ###########
C#####################################################################
```

```
      SUBROUTINE INTERP
      COMMON /INT/ YN(7,7),SN(7,7),COEF(4),SNOD(12,7),YNOD(12,7)
C#######################################################################
C#####   FILL IN BOUNDARY NODES FOR CYLINDER FUNCTION        ###########
C#######################################################################
C     WRITE(11,1011) (COEF(K), K=1, 4)
 1011 FORMAT(' ', ' COEFFICIENTS'/, 4F10.5)
      DO 10 J=3,7
      YN(1,J)=YN(5,J)
      YN(6,J)=YN(2,J)
      SN(1,J)=SN(5,J)
      SN(6,J)=SN(2,J)
      YN(7,J)=YN(3,J)
      SN(7,J)=SN(3,J)
   10 CONTINUE
C#######################################################################
C#####          INTERPOLATE TO FILL IN GRID                  ###########
C#######################################################################
      KNT=0
      DO 25 I=1,10,3
      DO 20 J=3,7
      YNOD(I+1,J)=0.0
      YNOD(I+2,J)=0.0
      SNOD(I+1,J)=0.0
      SNOD(I+2,J)=0.0
      VAR1=0.0
      VAR11=0.0
      VAR2=0.0
      DO 15 K=1,4
      YNOD(I+1,J)=YNOD(I+1,J)+COEF(K)*YN(K+KNT,J)
      YNOD(I+2,J)=YNOD(I+2,J)+COEF(K)*YN(5-K+KNT,J)
      VAR1=VAR1+COEF(K)*COEF(K)*SN(K+KNT,J)*SN(K+KNT,J)
      VAR2=VAR2+COEF(K)*COEF(K)
      COEF2=COEF(K)*COEF(K)
      SNOD2=SN(5-K+KNT,J)*SN(5-K+KNT,J)
      VAR11=VAR11+COEF2*SNOD2
   15 CONTINUE
      SNOD(I+1,J)=SQRT(VAR1/VAR2)
      SNOD(I+2,J)=SQRT(VAR11/VAR2)
   20 CONTINUE
      KNT=KNT+1
   25 CONTINUE
      DO 30 I=1,12
      YNOD(I,1)=YNOD(I,3)
      SNOD(I,1)=SNOD(I,3)
      YNOD(I,2)=0.0
      SNOD(I,2)=0.0
   30 CONTINUE
C     WRITE(11,1001)
 1001 FORMAT(' ', ' NODAL VALUES '/)
C     DO 35 I=1, 12
C     WRITE(11,1007) (YNOD(I,J), J=1, 7)
   35 CONTINUE
 1007 FORMAT(' ', 8F10.5)
C     WRITE(11,1003)
```

```
 1003 FORMAT(' ', '  NODAL STD DEV '/)
C     DO 40 I=1, 12
C     WRITE(11,1007) (SNOD(I,J), J=1, 7)
   40 CONTINUE
      RETURN
      END
C######################################################################
C#####   SUBROUTINE RAND(DRAW,IENT)                          ###########
C#####   BUILD 10X10 TABLE OF RANDOM NUMBERS AND DRAW        ###########
C#####   FROM TABLE WITH REPLACEMENT.                        ###########
C######################################################################
      SUBROUTINE RAND(DRAW,IENT)
      DIMENSION TAB(10,10)
      COMMON /RNN/ NR, N1, N2
      IF(IENT.NE.1) GO TO 1003
      CALL RANDO(XRN)
      II=INT(10.0*XRN)+1
      CALL RANDO(XRN)
      JJ=INT(10.0*XRN)+1
      DO 1000 I=1,10
      DO 1001 J=1,10
      CALL RANDO(XRN)
 1001 TAB(I,J)=XRN
 1000 CONTINUE
C     DO 1007 I=1, 10
C     WRITE(10,1111) (TAB(I,J), J=1, 10)
 1111 FORMAT(' ', 10F8.5)
 1007 CONTINUE
      IC=1
 1003 DRAW=TAB(II,JJ)
      CALL RANDO(XRN)
      TAB(II,JJ)=XRN
      IF(MOD(IC,2).EQ.0) GO TO 1005
      CALL RANDO(XRN)
      II=INT(10.0*XRN)+1
      IC=2
      GO TO 1006
 1005 CALL RANDO(XRN)
      JJ=INT(10.0*XRN)+1
      IC=1
 1006 CONTINUE
      RETURN
      END
C######################################################################
C#####   SUBROUTINE RANDO                                    ###########
C#####   GENERATE UNIFORM RANDOM NUMBERS                     ###########
C######################################################################
      SUBROUTINE RANDO(R)
      COMMON /RNN/ IX, IY, IZ
      IX = 171 * MOD(IX,177) - 2 * (IX/177)
      IY = 172 * MOD(IY,176) - 35 * (IY/176)
      IZ = 170 * MOD(IZ,178) - 63 * (IZ/178)
      IF (IX .LT. 0) IX = IX + 30269
      IF (IY .LT. 0) IY = IY + 30307
      IF (IZ .LT. 0) IZ = IZ + 30323
```

```
      R=AMOD(FLOAT(IX)/30269.+FLOAT(IY)/30307.+FLOAT(IZ)/30323,1.0)
      RETURN
      END
C######################################################################
C#####   SUBROUTINE RANDN(RANUM)                            ###########
C#####   APPROXIMATE RANDOM NORMAL NUMBERS                  ###########
C######################################################################
      SUBROUTINE RANDN(RANUM)
      COMMON /RNN/ NR, N1, N2
      RANUM=0.
      DO 10 I=1,12
      CALL RAND(X,2)
      RANUM=RANUM+X
   10 CONTINUE
      RANUM=RANUM-6.
C     WRITE(10,1001) RANUM
 1001 FORMAT(' ', F10.4)
      RETURN
      END
C######################################################################
C#####   SUBROUTINE EICAL(K)                                ###########
C#####   MONTHLY ESTIMATES OF EI FROM SEASONALLY            ###########
C#####   CONTINUOUS PROBABILITY DISTRIBUTIONS.              ###########
C######################################################################
      SUBROUTINE EICAL(K)
      COMMON /INT/ YN(7,7),SN(7,7),COEF(4),SNOD(12,7),YNOD(12,7)
      COMMON /EI/ HSK(12), RNOD(12,7), HB(12), HSD(12), HMIN(12),
     *    CP(12), VR(12), F1(12), D1(12), C(4), HC
      COMMON /RNN/ NR, N1, N2
      VS = 1.
      CALL RAND(R,2)
C######################################################################
C#####   LOCATE RANDOM NUMBER ON PROBABILITY FUNCTION       ###########
C#####   BY INTERVAL HALVING.                               ###########
C######################################################################
      IF (R.LT.RNOD(K,3)) GO TO 105
      IF (R.LT.RNOD(K,4)) GO TO 110
      IF (R.LT.RNOD(K,5)) GO TO 115
      IF (R.LT.RNOD(K,6)) GO TO 120
      HC = HMIN(K)
      GO TO 155
  105 BL=0.0
      BR=VS
      IS=1
      GO TO 125
  110 BL=VS
      BR=2.0*VS
      IS=2
      GO TO 125
  115 BL=2.0*VS
      BR=3.0*VS
      IS=3
      GO TO 125
  120 BL=3.0*VS
      BR=4.0*VS
```

```
      IS=4
  125 A=(-BL)/(BR-BL)-0.5
      C(1)=(9.*(RNOD(K,IS+1)+RNOD(K,IS+2))-RNOD(K,IS)-RNOD(K,IS+3))/16.
      C(2)=(11.*(RNOD(K,IS+2)-RNOD(K,IS+1))+RNOD(K,IS)-RNOD(K,IS+3))/8.
      C(3)=(RNOD(K,IS)-RNOD(K,IS+1)-RNOD(K,IS+2)+RNOD(K,IS+3))/4.
      C(4)=(3.*(RNOD(K,IS+1)-RNOD(K,IS+2))-RNOD(K,IS)+RNOD(K,IS+3))/2.
      B=1./(BR-BL)
  130 Z=A+B*(BL+BR)/2.0
      PC=((C(4)*Z+C(3))*Z+C(2))*Z+C(1)
      IF(R.LT.PC) GO TO  135
      BL=(Z-A)/B
      GO TO  140
  135 BR=(Z-A)/B
  140 IF(ABS(BR-BL).LT.0.001) GO TO  145
      GO TO  130
  145 V=(BL+BR)/2.
      IF(V.LE.CP(K)) GO TO  150
C#######################################################################
C#####   COMPUTE EI AT RANDOM PROBABILITY                    ###########
C#######################################################################
      HC=HSD(K)*(ALOG((V-VR(K))/(CP(K)-VR(K)))/F1(K)-HSK(K)/8)+HB(K)
      GO TO  155
  150 CONTINUE
      HC=-HSD(K)*(ALOG(V/CP(K))/D1(K)+HSK(K)/8)+HB(K)
  155 CONTINUE
C#######################################################################
C#####   PULSE NODES WITH RANDOM NORMAL DEVIATE              ###########
C#######################################################################
      DO  160 L=3,7
      CALL RANDN(RN)
      RNOD(K,L)=YNOD(K,L)+RN*SNOD(K,L)
  160 CONTINUE
      RNOD(K,1)=RNOD(K,3)
C     WRITE(8,1001) K,HC,R
 1001 FORMAT(' ', I5,2F10.3)
      RETURN
      END
C#######################################################################
C#####   SUBROUTINE CFAC                                     ###########
C#####   ESTIMATES OF CROPPING FACTOR FROM DOUBLE            ###########
C#####   TRIANGLE DISTRIBUTION.                              ###########
C#######################################################################
      SUBROUTINE CFAC(K)
      DIMENSION ACOEF1(4), ACOEF2(4), BCOEF(4), A(4), B(4)
      DIMENSION C(5), P(4), TA(4)
      COMMON /CRP/ CF(24), CC
      COMMON /RNN/ NR, N1, N2
C#######################################################################
C#####   INITIALIZE COEFFICIENTS FOR DISTRIBUTION            ###########
C#######################################################################
      IF (INIT.EQ.1) GO TO 5
      INIT = 1
      DATA ACOEF1 /0.0, 0.179, 1.0, 0.179/
      DATA ACOEF2 /0.179, 0.821, -0.821, -0.179/
      DATA BCOEF /0.179, 0.821, -0.821, -0.179/
```

```
    5 CONTINUE
C######################################################################
C#####   CHOOSE HALF-MONTH C ESTIMATE                       ###########
C######################################################################
      CALL RAND(R1,2)
      JJ = 1
      IF (R1.LE.0.5) JJ = 0
C######################################################################
C#####   CALCULATE DISTRIBUTION COEFFICIENTS                ###########
C######################################################################
      C(3) = CF(2*K-JJ)
      C(1) = 0.9 * C(3)
      C(5) = 1.1 * C(3)
      IF (C(5).GT.1.0) C(5) = 1.0
      C(2) = (C(1)+C(3))/2.
      C(4) = (C(3)+C(5))/2.
      DO 10 I=1, 4
      A(I) = ACOEF1(I) - ACOEF2(I)*C(I)/(C(I+1)-C(I))
      B(I) = BCOEF(I)/(C(I+1)-C(I))
   10 CONTINUE
      TAREA = 0.0
      DO 15 I=1, 4
      AAI = A(I)*(C(I+1)-C(I)) + B(I)*(C(I+1)*C(I+1)-C(I)*C(I))/2.
      TA(I) = TAREA
      P(I) = TAREA + AAI
      TAREA = P(I)
   15 CONTINUE
      DO 20 I=1, 4
      P(I) = P(I)/TAREA
   20 CONTINUE
C######################################################################
C#####   OBTAIN ESTIMATE FROM DISTRIBUTION                  ###########
C######################################################################
      CALL RAND(R,2)
      IB=1
      IF (R.GT.P(1)) IB = 2
      IF (R.GT.P(2)) IB = 3
      IF (R.GT.P(3)) IB = 4
      BL = C(IB)
      BR = C(IB+1)
      A1 = (-BL)/(BR-BL) - 0.5
      B1 = 1./(BR-BL)
   25 Z = A1 + B1*(BL+BR)/2.
      CC = (BR+BL)/2.
      RP = (TA(IB)+A(IB)*(CC-C(IB))+B(IB)*(CC*CC-C(IB)*C(IB))/2.)/TAREA
      IF (R.LT.RP) GO TO 30
      BL = (Z-A1)/B1
      GO TO 40
   30 BR = (Z-A1)/B1
   40 IF (ABS(BR-BL).LE.0.0001) GO TO 50
      GO TO 25
   50 CONTINUE
C     WRITE(9,1001) K,CC,R
 1001 FORMAT(I5, 2F10.3)
      RETURN
      END
```

Program Output

Mono Crop Soybeans Sp Turn Conv Till Fair prod Late season

RANDOM NUMBER SEEDS			NI	NP
15418	7324	28483	100	10

MONTH	MEAN	STD DEV	SKEW	HMIN
1	31.946	32.920	2.674	.000
2	25.589	23.966	1.808	.000
3	43.949	38.429	1.393	.000
4	47.878	48.464	1.641	.000
5	54.000	46.069	1.209	.000
6	57.330	63.216	1.854	.000
7	117.226	114.490	1.527	.000
8	68.719	75.883	1.768	.000
9	42.958	44.973	2.198	.000
10	25.911	27.991	1.917	.000
11	25.892	32.533	2.411	.000
12	23.347	25.074	2.196	.000

NODE	STD DEV
.124	.006
.234	.004
.520	.006
1.030	.008
1.645	.008
.229	.005
.357	.004
.511	.007
.984	.009
2.055	.009
.178	.007
.323	.012
.511	.008
1.026	.010
2.781	.010
.158	.007
.314	.009
.524	.007
.972	.014
2.304	.014

CROPPING FACTORS

.44	.44	.44	.44	.44	.44	.56	.56
.56	.56	.56	.83	.83	.70	.54	.54
.40	.40	.31	.26	.44	.44	.44	.44

HP	CP	VR	F	D
-.6361	3.0029	5.0529	1.0474	.7150
-.8417	2.6781	4.7281	1.2299	.9414
-.9695	2.5224	4.5724	1.3159	1.0695
-.7828	2.6152	4.6652	1.4377	1.1270
-1.0210	2.4535	4.5035	1.3752	1.1490
-.6751	2.6953	4.7453	1.4987	1.1399
-.8330	2.5726	4.6226	1.4306	1.1400
-.6846	2.6629	4.7129	1.5428	1.1877
-.6805	2.8241	4.8741	1.2526	.9093
-.6861	2.7188	4.7688	1.4296	1.0779
-.4945	2.9042	4.9542	1.5465	1.0916
-.6566	2.8236	4.8736	1.2991	.9432

MONTHLY TOTALS

MONTH 1

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	99.42	56.47	50.57	42.73	27.23	17.87	17.19	11.92	9.45	7.55
2	95.17	51.39	43.96	32.47	19.88	17.80	13.53	11.68	8.92	5.93
3	93.73	50.52	38.35	31.35	19.83	15.07	12.08	10.83	8.71	5.85
4	90.74	49.06	35.59	28.83	19.55	14.56	11.95	10.19	8.53	5.81
5	74.10	46.57	35.46	28.68	18.27	13.74	11.58	9.74	8.48	5.73
(Abbreviated)										
96	19.35	12.19	9.16	7.58	6.09	5.24	3.60	2.73	1.59	.77
97	17.75	11.88	9.03	7.28	6.02	4.96	3.58	2.68	1.48	.70
98	16.77	10.73	8.04	7.08	5.78	4.13	3.55	2.32	1.33	.63
99	15.06	10.22	7.72	6.65	5.71	4.10	3.38	2.07	1.26	.61
100	12.75	7.82	7.31	5.00	4.37	3.69	3.34	1.69	1.10	.59

MONTH 2

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	46.69	34.16	28.39	27.06	20.14	11.31	10.52	8.14	7.21	7.02
2	42.02	33.30	26.90	25.59	19.52	11.05	9.93	8.02	6.95	5.75
3	41.35	29.54	26.81	23.08	17.03	10.95	8.80	7.92	6.90	5.11
4	39.89	29.40	25.19	21.50	14.41	10.55	8.74	7.76	6.17	4.71
5	39.80	29.20	25.15	21.49	13.65	10.44	8.39	7.53	5.94	4.62
(Abbreviated)										
96	12.27	9.67	8.34	6.52	5.41	3.89	3.15	1.69	.80	.36
97	12.12	9.43	8.22	6.49	5.37	3.66	3.02	1.43	.79	.34
98	11.30	9.01	8.03	6.47	4.79	3.56	2.22	1.27	.78	.27
99	11.29	8.85	7.48	5.61	4.61	3.55	1.83	.86	.70	.22
100	10.36	8.08	6.82	4.70	4.46	2.93	1.62	.81	.60	.09

MONTH 3

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	76.70	61.15	48.36	39.61	34.11	32.44	20.23	14.80	11.02	8.88
2	71.09	52.55	46.80	35.91	30.93	28.77	19.68	13.54	10.83	8.85
3	69.19	50.56	43.51	35.89	29.89	23.59	18.52	13.42	10.75	8.35
4	68.74	49.98	42.66	35.64	29.65	23.56	15.30	13.00	10.43	8.12
5	65.59	49.87	39.85	34.11	29.54	20.07	14.92	12.14	10.29	8.03
				(Abbreviated)						
96	25.20	18.38	14.65	11.74	9.42	7.66	6.22	3.47	1.17	.13
97	22.45	18.26	14.43	11.63	9.32	7.27	5.59	3.32	1.07	.08
98	22.07	17.97	14.20	11.42	8.93	7.17	5.37	3.29	1.03	.05
99	18.99	14.76	13.58	10.93	8.05	6.32	4.47	2.69	.88	.01
100	18.49	13.80	13.39	9.14	8.01	4.40	4.37	1.84	.61	.00

MONTH 4

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	132.56	81.21	74.85	60.26	48.65	40.36	28.12	21.85	18.71	14.59
2	116.41	78.08	70.78	58.91	47.48	40.15	26.66	19.48	15.03	10.90
3	111.57	77.80	63.60	58.80	47.41	38.07	24.08	18.48	15.01	10.04
4	109.69	76.52	63.41	54.92	41.98	32.66	22.36	16.64	14.16	8.74
5	101.99	72.49	62.86	54.01	41.58	31.18	21.61	16.53	12.35	7.76
				(Abbreviated)						
96	31.95	23.79	16.25	13.42	9.37	6.85	5.23	2.67	.82	.00
97	30.52	21.28	15.92	11.90	9.00	6.83	4.43	2.23	.75	.00
98	27.87	21.09	15.00	11.44	7.48	5.98	3.10	1.91	.65	.00
99	26.34	18.82	14.52	11.07	6.11	5.13	2.37	1.84	.55	.00
100	18.22	17.84	11.97	10.61	5.03	3.44	2.29	1.33	.49	.00

MONTH 5

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	111.21	82.65	70.09	56.34	54.39	41.95	36.63	27.92	20.23	16.54
2	106.84	82.44	63.01	56.29	50.99	40.76	34.93	23.76	20.02	15.45
3	105.86	78.75	62.93	54.80	47.47	38.86	30.72	23.63	18.70	13.63
4	105.44	78.03	62.37	53.82	47.25	37.83	29.14	22.27	18.17	12.36
5	102.70	74.09	59.47	51.63	46.99	37.08	27.13	21.77	17.32	12.32
				(Abbreviated)						
96	45.07	30.29	23.91	18.08	14.35	9.93	6.49	2.93	.97	.00
97	37.37	29.98	22.41	17.99	13.96	9.92	6.33	2.79	.90	.00
98	36.86	29.58	21.32	16.60	12.10	9.61	5.68	2.61	.73	.00
99	35.63	26.77	20.65	16.05	11.25	9.27	5.43	2.58	.35	.00
100	35.30	22.83	16.94	14.43	10.62	9.10	5.05	2.12	.06	.00

MONTH 6

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	255.29	159.35	118.60	84.23	79.22	54.33	52.39	41.55	24.49	12.96
2	247.64	138.29	97.27	80.35	76.00	50.00	46.39	36.96	16.38	11.05
3	208.91	137.50	90.93	79.78	67.88	49.05	34.93	30.56	16.21	9.44
4	201.43	130.50	90.55	76.62	61.83	48.25	34.53	28.93	16.08	9.29
5	191.15	122.04	89.46	75.47	61.25	47.51	33.12	27.91	15.70	9.01
				(Abbreviated)						
96	46.01	34.55	23.83	18.44	13.40	9.36	7.42	4.20	1.67	.67
97	44.90	33.95	23.80	18.21	12.78	9.26	5.96	4.20	1.66	.63
98	40.10	30.46	22.22	16.65	11.40	7.86	4.83	3.88	1.52	.50
99	34.61	28.20	19.71	14.68	11.19	7.30	4.55	3.70	1.35	.39
100	22.04	17.37	13.24	11.53	8.89	5.91	4.29	2.16	.59	.17

MONTH 7

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	363.67	286.94	261.40	216.65	159.57	118.00	90.41	83.39	59.36	57.65
2	357.34	274.88	230.19	191.41	153.71	102.01	75.08	64.28	53.15	40.03
3	346.18	261.60	227.53	175.02	142.83	99.95	73.95	64.22	47.62	35.74
4	335.47	258.44	223.28	172.31	141.29	98.81	73.57	60.47	45.72	32.73
5	328.12	249.66	207.66	171.80	126.78	95.13	73.41	57.93	45.44	27.53
				(Abbreviated)						
96	106.70	68.11	59.52	42.26	32.20	25.18	13.40	9.53	4.35	2.06
97	101.31	67.86	58.70	41.78	29.06	20.03	11.33	8.28	4.18	1.99
98	95.00	64.78	57.59	36.83	28.14	18.88	11.29	7.45	4.15	1.82
99	84.06	64.76	57.48	33.15	26.85	14.39	10.99	7.36	4.12	1.81
100	82.69	62.76	33.57	32.88	17.55	12.10	9.72	6.25	3.10	1.09

MONTH 8

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	153.44	103.55	82.60	72.52	71.61	54.16	47.08	34.11	26.22	21.63
2	147.91	103.07	81.93	72.20	67.75	51.36	33.58	29.70	24.54	13.43
3	147.58	102.59	77.92	71.74	58.42	40.54	33.57	29.12	23.69	13.22
4	144.80	100.46	76.67	71.38	54.33	39.23	32.74	28.47	23.50	10.16
5	141.21	99.49	75.14	68.85	49.67	39.05	29.62	27.70	16.53	9.01
				(Abbreviated)						
96	40.98	27.41	20.03	13.76	8.98	5.33	4.13	3.02	1.59	.29
97	40.97	27.31	19.56	10.16	7.23	5.25	3.89	2.56	1.40	.27
98	38.19	26.22	17.93	9.98	6.25	5.19	3.55	2.01	.97	.20
99	37.78	24.29	17.35	9.76	5.62	5.15	3.23	1.88	.84	.09
100	28.08	19.70	12.08	7.49	5.14	5.11	2.37	.76	.69	.00

MONTH 9

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	94.18	61.38	55.59	52.36	40.21	38.66	25.61	24.11	11.74	5.88
2	87.59	57.51	53.96	41.24	38.75	27.20	20.42	12.05	8.30	5.64
3	85.15	54.77	50.38	40.65	35.48	22.29	15.08	11.78	8.22	5.36
4	72.06	54.29	47.99	39.61	30.96	21.62	14.65	11.20	7.58	4.90
5	69.81	53.35	43.67	39.21	28.16	20.00	13.97	11.14	7.25	4.88
				(Abbreviated)						
96	22.77	16.98	12.07	8.49	4.41	3.65	2.42	1.06	.25	.00
97	22.06	16.31	11.66	8.06	4.28	3.62	2.28	.64	.22	.00
98	22.01	15.89	11.28	7.25	4.28	3.54	2.28	.50	.16	.00
99	21.33	14.43	11.20	6.83	3.87	2.81	1.88	.49	.14	.00
100	19.50	14.33	10.79	6.55	3.25	2.33	1.17	.35	.00	.00

MONTH 10

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	31.88	21.52	20.89	17.60	14.50	13.21	9.97	7.38	4.27	2.58
2	30.13	21.52	19.27	17.41	13.51	10.90	9.75	6.98	3.81	2.09
3	29.28	21.50	18.34	14.68	13.44	10.04	8.88	4.92	3.67	1.80
4	28.88	21.38	18.06	14.56	12.04	10.03	7.47	4.87	3.36	1.71
5	27.84	20.32	18.03	14.18	11.51	9.86	6.44	4.45	3.14	1.48
				(Abbreviated)						
96	8.81	5.78	4.26	3.62	2.62	1.94	1.28	.48	.14	.00
97	8.78	5.70	3.83	3.29	2.54	1.84	1.14	.48	.14	.00
98	8.02	5.55	3.69	3.22	2.28	1.83	1.11	.47	.13	.00
99	6.46	4.78	3.67	3.16	2.21	1.69	.88	.42	.06	.00
100	6.02	3.42	2.82	2.36	2.09	1.02	.79	.33	.03	.00

MONTH 11

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	58.23	38.18	28.84	22.83	17.79	15.70	11.83	8.74	6.72	5.20
2	56.12	35.95	26.73	21.21	17.21	15.46	10.54	8.12	6.41	4.88
3	52.83	35.31	25.52	20.12	15.97	13.07	10.10	7.54	6.27	4.22
4	52.65	31.81	25.30	19.99	15.81	12.98	9.25	7.34	5.69	3.61
5	49.27	31.50	24.25	17.87	15.59	12.27	8.55	7.24	5.61	3.05
				(Abbreviated)						
96	13.69	9.06	6.93	6.02	4.06	3.33	2.17	1.11	.31	.00
97	13.15	8.20	6.39	5.30	4.02	2.73	1.98	1.05	.29	.00
98	12.48	8.07	6.35	4.82	3.91	2.35	1.64	.72	.05	.00
99	8.48	7.87	6.02	4.29	3.49	2.03	1.38	.71	.03	.00
100	7.73	7.36	3.29	2.04	1.66	1.16	1.16	.69	.00	.00

MONTH 12

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	48.22	39.41	36.06	22.47	16.85	12.32	10.52	8.62	6.20	5.73
2	47.18	38.50	26.07	21.83	14.57	11.16	9.55	7.29	5.69	3.73
3	46.95	35.05	23.52	19.26	12.66	11.05	8.53	7.02	5.67	3.54
4	45.18	34.89	22.97	15.49	12.26	10.71	8.43	6.52	5.63	3.47
5	43.51	33.70	20.82	14.63	12.06	9.91	7.60	6.51	5.21	3.30
				(Abbreviated)						
96	11.00	8.52	7.26	6.16	4.51	3.43	2.68	1.08	.77	.26
97	10.93	7.79	6.99	6.05	4.38	3.33	2.66	.94	.72	.26
98	10.54	7.46	6.69	5.90	4.06	3.15	2.34	.89	.69	.23
99	9.52	7.26	6.52	5.45	3.83	2.76	1.16	.83	.37	.21
100	9.43	7.23	6.02	3.22	2.30	2.12	.98	.62	.36	.16

YEARLY TOTALS

SORTED DATA

	1	2	3	4	5	6	7	8	9	10
1	626.67	596.44	495.31	418.60	393.58	365.01	319.74	288.61	267.26	264.99
2	625.33	543.02	465.56	406.90	382.48	337.44	300.84	277.80	254.81	248.65
3	618.09	512.35	463.04	393.37	376.05	326.38	295.41	273.01	254.38	232.03
4	612.36	493.07	448.06	388.29	365.96	323.33	294.27	272.76	252.48	231.93
5	596.68	490.44	430.82	383.35	359.41	314.62	294.02	271.65	249.65	222.45
				(Abbreviated)						
96	319.07	275.22	251.57	234.56	220.09	202.02	188.05	162.47	148.74	106.34
97	318.89	271.73	240.36	233.29	217.56	201.26	180.52	162.17	144.71	102.29
98	299.48	250.74	237.10	228.74	216.45	199.41	172.33	159.97	138.55	98.18
99	284.01	246.26	231.29	223.65	215.48	199.03	169.34	159.90	125.01	82.91
100	282.55	229.54	219.91	219.50	191.09	181.28	169.19	158.89	124.06	82.77